Fourier transform examples

The box function and its Fourier transform

In[151]:=
```
fp = FourierParameters → {0, -2 Pi}
```
Out[151]=
```
FourierParameters → {0, -2 π}
```
In[152]:=
```
$Assumptions = {a > 0, b > 0}
```
Out[152]=
```
{a > 0, b > 0}
```
In[153]:=
```
PiecewiseExpand[UnitStep[t - a]]
```
Out[153]=
```
{ 1  a - t ≤ 0
{ 0  True
```
Defining the box function with width 2a and height 1

In[154]:=
```
f[t_, a_] = UnitStep[t + a] - UnitStep[t - a]
```
Out[154]=
```
-UnitStep[-a + t] + UnitStep[a + t]
```
In[155]:=
```
fpw[t_, a_] = PiecewiseExpand[f[t, a]]
```
Out[155]=
```
{ -1  a - t ≤ 0 && a + t < 0
{ 1   a - t > 0 && a + t ≥ 0
{ 0   True
```
In[156]:=
```
Plot[fpw[t, 3], {t, -5, 5}, Filling → Axis]
```
Out[156]=



The norm of the box function

In[157]:=
```
norm2f[a_] = Sqrt[Integrate[fpw[s, a]^2, {s, -Infinity, Infinity}]]
```
Out[157]= $\sqrt{2}\ \sqrt{a}$

The normalized box function

In[158]:=
```
ff[t_, a_] = fpw[t, a] / norm2f[a]
```
Out[158]=
```
{ -1  a - t ≤ 0 && a + t < 0
{ 1   a - t > 0 && a + t ≥ 0    / (√2 √a)
{ 0   True
```
Displaying the normalized box function

In[159]:=
```
Manipulate[
 Plot[ff[t, a], {t, -5, 5}, PlotRange -> {-1, 2}, Filling → Axis], {a, 1, 3}]
```

Out[159]=



The Fourier transform of the normalized box function

In[160]:=
```
fhat[s_, a_] = FourierTransform[ff[t, a], t, s, fp]
```

Out[160]= $\frac{\mathrm{Sin}[2\, a\, \pi\, s]}{\sqrt{2}\, \sqrt{a}\, \pi\, s}$

Checking that the Fourier transform is again normalized

In[161]:=
```
Integrate[fhat[s, a]^2, {s, -Infinity, Infinity}, Assumptions → a > 0]
```

Out[161]= 1

Comparing the normalized box function and its Fourier transform when changing the width parameter a

In[162]:=
```
Manipulate[Plot[{f[x, a] / norm2f[a], fhat[x, a]},
  {x, -5, 5}, PlotRange -> {-1, 1}, Filling → Axis], {a, 1, 3}]
```

Out[162]=



The hat function and its Fourier transform

Defining the hat function of width 2a (as a convolution of the box function with itself)

In[163]:=
```
g[t_, a_] = 1 / (2 * a) * Integrate[f[x, a] * f[2 * t - x, a], {x, -Infinity, Infinity}]
```

Out[163]= $\frac{1}{a}\left(-2\,t\,\text{UnitStep}[t] + (a+t)\,\text{UnitStep}[a+t] + (-a+t)\,\text{UnitStep}[-2\,a+2\,t]\right)$

In[164]:=
```
Manipulate[
 Plot[g[t, a] / (2 * a), {t, -5, 5}, PlotRange -> {-1, 1}, Filling → Axis], {a, 1, 3}]
```

Out[164]=



The norm of the hat function

In[165]:=
```
norm2g[a] = Sqrt[Integrate[PiecewiseExpand[g[t, a]]^2, {t, -Infinity, Infinity}]]
```

Out[165]= $\sqrt{\frac{2}{3}}\,\sqrt{a}$

The normalized hat function

In[166]:=
```
gg[t_, a_] = g[t, a] / norm2g[a]
```

Out[166]= $\frac{1}{a^{3/2}}\sqrt{\frac{3}{2}}\left(-2\,t\,\text{UnitStep}[t] + (a+t)\,\text{UnitStep}[a+t] + (-a+t)\,\text{UnitStep}[-2\,a+2\,t]\right)$

... write as a piecwise defined function

In[167]:=
```
PiecewiseExpand[g[t, a], a > 0]
```

Out[167]=

$$\begin{cases} \frac{-a-t}{a} & t \geq 0 \;\&\&\; a+t < 0 \;\&\&\; a-t \leq 0 \\ \frac{a-t}{a} & t \geq 0 \;\&\&\; a+t \geq 0 \;\&\&\; a-t > 0 \\ -\frac{2\,t}{a} & t \geq 0 \;\&\&\; a+t < 0 \;\&\&\; a-t > 0 \\ \frac{2\,t}{a} & t < 0 \;\&\&\; a+t \geq 0 \;\&\&\; a-t \leq 0 \\ \frac{-a+t}{a} & t < 0 \;\&\&\; a+t < 0 \;\&\&\; a-t \leq 0 \\ \frac{a+t}{a} & t < 0 \;\&\&\; a+t \geq 0 \;\&\&\; a-t > 0 \\ 0 & \text{True} \end{cases}$$

The Fourier transform of the normalized hat function

In[168]:=
```
FourierTransform[gg[t, a], t, s, fp]
```

Out[168]=
$$\frac{\sqrt{\frac{3}{2}}}{2\, a^{3/2}\, \pi^2\, s^2} - \frac{\sqrt{\frac{3}{2}}\, e^{-2\, i\, a\, \pi\, s}}{4\, a^{3/2}\, \pi^2\, s^2} - \frac{\sqrt{\frac{3}{2}}\, e^{2\, i\, a\, \pi\, s}}{4\, a^{3/2}\, \pi^2\, s^2}$$

... equivalently

In[169]:=
```
ghat[s_, a_] = ExpToTrig[%]
```

Out[169]=
$$\frac{\sqrt{\frac{3}{2}}}{2\, a^{3/2}\, \pi^2\, s^2} - \frac{\sqrt{\frac{3}{2}}\, \text{Cos}[2\, a\, \pi\, s]}{2\, a^{3/2}\, \pi^2\, s^2}$$

The Fourier transform is again normalized

In[170]:=
```
Integrate[ghat[s, a]^2, {s, -Infinity, Infinity}]
```

Out[170]= 1

Comparing the normalized hat function and its Fourier transform when changing the width parameter a

In[171]:=
```
Manipulate[Plot[{gg[t, a], ghat[t, a]},
  {t, -5, 5}, PlotRange -> {-1, 2}, Filling → Axis], {a, 1, 3}]
```

Out[171]=



A parametrized function with exponential decay

In[172]:= `h[t_, a_] = Exp[-a * Abs[t]]`

Out[172]= $e^{-a\,\text{Abs}[t]}$

In[173]:= `Manipulate[Plot[h[t, a], {t, -2, 2}, Filling → Axis], {a, 1, 3}]`

Out[173]=



The norm of this function

In[174]:= `norm2h[a_] = Sqrt[Integrate[h[t, a]^2, {t, -Infinity, Infinity}]]`

Out[174]= $\sqrt{\frac{1}{a}}$

The normed version of this function

In[175]:= `hh[t_, a_] = h[t, a] / norm2h[a]`

Out[175]= $\frac{e^{-a\,\text{Abs}[t]}}{\sqrt{\frac{1}{a}}}$

Displaying the normed function

In[176]:=
```
Manipulate[Plot[hh[t, a], {t, -2, 2}, PlotRange → {0, 2}, Filling → Axis], {a, 1, 3}]
```

Out[176]=



The Fourier transform of this function

In[177]:=
```
hhat[s_, a_] = FourierTransform[hh[t, a], t, s, fp]
```

Out[177]=

$$\frac{2}{\left(\frac{1}{a}\right)^{3/2} \left(a^2 + 4\,\pi^2\,s^2\right)}$$

Comparing the normalized function and its Fourier transform when changing the decay parameter a

In[178]:=
```
Manipulate[Plot[{hh[t, a], hhat[t, a]},
  {t, -2, 2}, PlotRange → {0, 2}, Filling → Axis], {a, 1, 3}]
```

Out[178]=



A parametrized gaussian function

In[179]:=
```
k[t_, a_] = Exp[-a * t^2]
```

Out[179]= $e^{-a\,t^2}$

The norm of this function

In[180]:=
```
norm2k[a_] = Sqrt[Integrate[k[t, a]^2, {t, -Infinity, Infinity}]]
```

Out[180]= $\frac{\left(\frac{\pi}{2}\right)^{1/4}}{a^{1/4}}$

The normalized version of this function

In[181]:=
```
kk[t_, a_] = k[t, a] / norm2k[a]
```

Out[181]= $a^{1/4}\, e^{-a\,t^2} \left(\frac{2}{\pi}\right)^{1/4}$

In[182]:=
```
Integrate[kk[t, a]^2, {t, -Infinity, Infinity}]
```

Out[182]= 1

Displaying the normalized function

In[183]:=
```
Manipulate[Plot[kk[t, a], {t, -2, 2}, PlotRange → {0, 2}, Filling → Axis], {a, 1, 3}]
```

Out[183]=



The Fourier transform of the normalized function

In[184]:=
```
khat[s_, a_] = FourierTransform[kk[t, a], t, s, fp]
```

Out[184]= $\frac{e^{-\frac{\pi^2 s^2}{a}} (2\,\pi)^{1/4}}{a^{1/4}}$

Checking the norm of the Fourier transform

In[185]:=
```
Integrate[khat[t, a]^2, {t, -Infinity, Infinity}]
```
Out[185]= 1

Comparing the normalized function and its Fourier transform when changing the variance parameter a

In[186]:=
```
Manipulate[Plot[{kk[t, a], khat[t, a]},
  {t, -2, 2}, PlotRange → {0, 2}, Filling → Axis], {a, 0.1, 6}]
```
Out[186]=



Checking the uncertainty relation

for the box function

In[187]:=
```
Integrate[ff[t, a]^2 * t^2, {t, -Infinity, Infinity}]
```
Out[187]= $\frac{a^2}{3}$

In[188]:=
```
Integrate[fhat[s, a]^2 * s^2, {s, -Infinity, Infinity}]
```
Out[188]= $\int_{-\infty}^{\infty} \frac{\text{Sin}[2\,a\,\pi\,s]^2}{2\,a\,\pi^2}\,ds$

for the hat function

In[189]:=
```
Integrate[gg[t, a]^2 * t^2, {t, -Infinity, Infinity}]
```
Out[189]= $\frac{a^2}{10}$

In[190]:=
```
Integrate[ghat[s, a]^2 * s^2, {s, -Infinity, Infinity}]
```
Out[190]= $\frac{3}{4\,a^2\,\pi^2}$

for the function with exponential decay

In[191]:=
```
Integrate[hh[t, a]^2 * t^2, {t, -Infinity, Infinity}]
```

Out[191]= $\frac{1}{2\, a^2}$

In[192]:=
```
Integrate[hhat[s, a]^2 * s^2, {s, -Infinity, Infinity}]
```

Out[192]= $\frac{a^2}{4\, \pi^2}$

for the gaussian function

In[193]:=
```
Integrate[kk[t, a]^2 * t^2, {t, -Infinity, Infinity}]
```

Out[193]= $\frac{1}{4\, a}$

In[194]:=
```
Integrate[khat[s, a]^2 * s^2, {s, -Infinity, Infinity}]
```

Out[194]= $\frac{a}{4\, \pi^2}$